International Journal on Electrical Engineering and Informatics - Volume 6, Number 3, September 2014

A New Spice Model Based on the Theoretical Application of Operational Transconductance Amplifier in Didactic Situation Optimization

Mohammad Reza Modabbernia[1], Yekta Nazarpour[2], Seyedeh Shiva Nejati[3], and Syrus Rouhollahi Moghadam[4]

[1]Electrical Engineering Group, Technical and Vocational University, Rasht branch, Rasht, IRAN
[2]Computer Engineering Group, Azad Islamic University, Lahijan Branch, Lahijan, IRAN,
[3]Electrical Engineering Group, Sardar Jangal University, Rasht, IRAN,
[4]Humanity Science Group, Technical and Vocaational University, Rasht branch, Rasht, IRAN,
m_modabbernia@afr.ac.ir, nazarpour_1ta@yahoo.com, s.nejati@sjihe.ac.ir, Moghadam.steve34@gmail.com

Abstract: In this paper, the application of operational transconductance amplifier (OTA), in addition to comparing with the interior structures of the most important available integrated circuits, have been analyzed. In this regard, a model in Spice has been presented to facilitate the trend of education for the undergraduate based on OTA theoretical analysis. The present paper aims to offer a model for the OTA in Spice from which the theoretical features of the said amplifiers can be incorporated. This causes the application of the OTA integrated circuits, under the impression and contribution of computerized simulation, and the accuracy of the theoretical analyses to be instructed and to be studied respectively. At the end, to show the merit and competence of the proposed model, three practical circuits, whose results are compared to the theoretical values and simulation accuracy, are represented. The proposed model covers the high speed of simulation and appropriate numerical convergence.

Keywords: Operational transconductance amplifier; OTA; PSpice; Spice netlist; Simulation

1. Introduction

Education forms the essence and pivotal axis of Industrial, economical and social developments of the societies. Over the past few decades, the researchers, with respect to computerized advancements and internet formation, have represented a lot of new and novel solutions and methods to develop the resources, techniques and educational instruments [1]. The simulation application on Spice has well found its direction, amid electronics-bases sources, promoting and improving the issue of education and also basic concepts elaboration. In this respect, the authors dramatically utilize components modelling and integrated circuits in simulations [2-6]. Selecting a model among diverse types of the represented models for an offered component or IC depends upon the simulation aims. If the aim is to predict the circuit behavior before manufacturing, the elements models should be highly exact. This can be led into more intricacy of the model and makes the simulation more complicated. However, on the condition that the only target is the study of the component behavior performance, view point of a know-how system or application principles, the offered model should be designed simply to accelerate and hasten the trend of simulation making the simulation results comparable to those of the theoretical ones [7-8].

An operational transconductance amplifier (OTA) is basically a voltage-dependent current source whose gain can be adjusted by the control current. Unlikely to operational amplifiers (Opamp), OTA simulation and modelling receives trivial heed and only its application and practical circuits are considered important [5-6].The nonlinear model of OTA can be found in [9-11]. In [12], there is a sub-circuit bipolar transistor for OTA based on the model 17. An OTA linear model for filter simulation and a phase structure model of OTA have been

Received: February 6th, 2014. Accepted: September 27th, 2014

successively studied in [13] and [14]. As well, the IC manufactures, in some extent, have represented a sub-circuit in spice for their own OTAs [15].

In other words, the IC manufacturing companies have offered precise and exact various macro models for better exclusive performance of the operational amplifiers (Opamp) [16-18]. However, in electronic references, a simple model of Opamp has been continuously utilized for the circuits' simulation introduced [4-6]. This satisfies the issue related to the thyristor macro model. The thyristor model is intricate in PSpice and ICAP software [19] for the simulation of power electronic circuits; however, the simple model of thyristor is used for education in [20-22]. This reflects that the application of the complicated models of the components and ICs are not suitable for the under graduated students involved in electronic basic concepts, but it is needed to choose a model which can make a proper compatibility between the theoretical and practical concepts. In case the effects of noise and frequency and also the practical constrains of OTA are foregone, the following items can be considered for the propitious performance of the macro model in electronics undergraduate students education.

1. Preparing a proportionate balance between simplicity and exactness by the model
2. Enjoying a proper convergence while simulating
3. Showing the input and output impedance of the amplifier
4. Enjoying the non-linear relation between the amplifier input and output
5. Having a suitable comparative proportion with the internal circuits of the manufacturing companies

The following pattern has been carried out to instruct OTA application, modeling and its utilization in this paper. At first, the general structure of OTA and its basic block diagram are presented. Forgoing the case study of each element, it is believed that via a basic and comprehensive framework a tremendous number of circuits and identical elements can be demonstrated in the course of education. This can be led into the elevation and declination of the students understanding and fatigue respectively.

Accordingly, the OTA macro model is going to be introduced step by step based on the general structure and block diagram of the OTA. This step by step modelling enjoys a didactic significance from two aspects. First of all, it makes the students implicitly familiarized to spice modelling and its instructions. Secondly, the performance of the OTA configuring layers can be expressed in this regard. Finally under the assistance of the represented OTA spice macro model, the three applicable circuits given by the manufacturing companies [23-25] can be simulated to prepare the ground for a comparison between the simulation result and those stemmed from the circuits' analysis in ICs data sheets. This is accomplished to reflect the model competence and merit.

2. General Structure of an Operational Transconductance Amplifier (OTA)

Figure 1 shows the block diagram of an operational transconductance amplifier (OTA). Input differential state like all other operational amplifiers should differentiate input signals and make an equivalent current signal in its collector's outputs. Current source provides the required current for biasing of the differential state. This current is generated with respect to external controller factor; therefore, the designer can control it. This issue causes high flexibility in IC suitable for the designer.

Loads of 1, 2 and 3 are current sources which act as active loads. Supposing that the output currents of these sources are equal to that i_{c1} and i_{c2} can be found in the output active loads of 2 and 3 respectively. This causes the OTA output current to be equal to $i_o = i_{c2} - i_{c1}$.

If in the simplest shape ,active loads and current source are considered as a simple mirror current source, general structure of Figure 1 can be considered equal to Figure 2 ,that is a very basic circuit of an OTA(4).



Figure 1. The block diagram of an OTA



Figure 2. Basic circuit of an OTA

Because, in this paper, all the features of OTAs follow this pattern, and only the used current source type is different in them; therefore, circuit in Figure 2 is studied by detail and the relevant mathematic equations of input and output are obtained. These equations can be easily extended to other internal circuits because other current sources have a better performance than simple current source. By assuming an equal output current and reference current accompanied by big output impedance, it can have more accuracy.

If it is assumed all Figure 2 related transistors are the same and have big β coefficient, then it can be said that output and reference mirror source currents are approximately equal (this equality is more obvious for Wilson current source used in OTA). Q_3 and Q_4 transistors consist of the current source. Q_5 and Q_6 transistors are considered as active load 1. Q_7 and Q_8 are taken into account as active load 2, and Q_9 and $Q_{10}$are active load 3. Supposing output and the references of these current sources are equal, we will have:

$$i_4 = i_3 = I_{ABC} \tag{1}$$
$$i_6 = i_5 \Rightarrow i_6 = i_2 \tag{2}$$
$$i_8 = i_7 \Rightarrow i_8 = i_1 \tag{3}$$
$$i_{10} = i_6 = i_9 \Rightarrow i_{10} = i_2 \tag{4}$$

KCL in output node results in:

$$i_8 = i_{10} + i_o \Rightarrow i_o = i_8 - i_{10} \Rightarrow i_o = i_1 - i_2 \tag{5}$$

If g_m is defined as the ratio of output signal to the difference between input voltages signal, in order to calculate g_m, $(i_1 - i_2)$ should have a relation with $v_1 - v_2$. With respect to this equation for Q_1 and Q_2 transistors:

$$i_1 = I_s \; e^{\frac{v_{be1}}{V_T}} \tag{6}$$
$$i_2 = I_s \; e^{\frac{v_{be2}}{V_T}} \tag{7}$$

In above equations, $I_s$is reverse saturation current and V_T is thermal voltage. With regard to KCL in emitter node of $Q_1$and Q_2 transistors, we have:

$$i_1 + i_2 = i_4 \Rightarrow I_{ABC} = i_1 + i_2 \tag{8}$$

By filling Equation (8) with Equation (6) and Equation (7), we will have:

$$I_{ABC} = I_s \left(e^{\frac{v_{be1}}{V_T}} + e^{\frac{v_{be2}}{V_T}} \right) \tag{9}$$

or

$$I_s = \frac{I_{ABC}}{e^{\frac{v_{be1}}{V_T}} + e^{\frac{v_{be2}}{V_T}}} \tag{10}$$

so

$$i_1 = I_{ABC} \frac{e^{\frac{v_{be1}}{V_T}}}{e^{\frac{v_{be1}}{V_T}} + e^{\frac{v_{be2}}{V_T}}} \tag{11}$$

$$i_2 = I_{ABC} \frac{e^{\frac{v_{be2}}{V_T}}}{e^{\frac{v_{be1}}{V_T}} + e^{\frac{v_{be2}}{V_T}}} \tag{12}$$

so it can be said that:

$$i_o = i_1 - i_2 = I_{ABC}\,\frac{e^{\frac{v_{be1}}{V_T}} - e^{\frac{v_{be2}}{V_T}}}{e^{\frac{v_{be1}}{V_T}} + e^{\frac{v_{be2}}{V_T}}} \tag{13}$$

or

$$i_o = I_{ABC} \times \tanh\left(\frac{v_{be1} - v_{be2}}{2V_T}\right) = I_{ABC} \times \tanh\left(\frac{v_1 - v_2}{2V_T}\right) \tag{14}$$

If, according to Equation (14), i_o current changes curve is plotted based on $v_1 - v_2$ changes, it can be seen that in a very small range of $v_1 - v_2$ changes, the output and input connection is linear, so the system goes to saturation state very fast. OTA circuits obtain hyperbolic tangent function of $v_1 - v_2$; therefore, for a proper performance, it should be assumed that the difference between v_1 and v_2 is very small, i.e. assuming $v_1 - v_2$ is very small $\left((v_1 - v_2) << 2V_T = 50\,\text{mV}\right)$, that is approximately $(v_1 - v_2) << 50\,\text{mV}$, we will have:

$$g_m = \left|\frac{\partial i_o}{\partial (v_1 - v_2)}\right| \tag{15}$$



Figure 3. Transfer characteristic curve of i_o per $v_1 - v_2$

$$\tanh\left(\frac{v_1 - v_2}{2V_T}\right) \approx \frac{v_1 - v_2}{2V_T} \tag{16}$$

$$g_m = \frac{I_{ABC}}{2V_T} \tag{17}$$

In Equation (17), V_T is constant and I_{ABC} reference current is determined by additional circuit (an adjustable circuit or a combination of voltage source and variable resistor). Therefore g_m value and thus transfer conductance can be controlled from outside. Voltage gain can be determined as below:

$$A_V = \frac{v_o}{v_1 - v_2} = \frac{i_o\ R_L}{v_1 - v_2} = g_m\ R_L = \frac{I_{ABC}\ R_L}{2V_T} \tag{18}$$

Voltage gain can be controlled by I_{ABC}.

3. Internal circuit of LM13700

LM13700 is an improved form of LM13600 that contains 2 transfer conductance operational amplifiers. Each one has 2 differential voltage inputs and 1 push pull output. Supply line of two amplifiers is common and other parts of them are individual. Figure 4 indicates internal circuits of these amplifiers.

1. Current source and active loads 1, 2, 3 of this integrated circuit have Wilson current source rather than simple mirror current source. The connection between the output current I_o and reference current in this source is:



Figure 4. Internal circuit of LM13700



Figure 5. Wilson current source in LM13700 structure

$$I_o = \frac{\beta^2 + 2\beta}{\beta^2 + 2\beta + 2} I_{ref} \qquad (19)$$

In Wilson current source even if β isn't very big, because it is in the power of 2, output current easily will become equal to reference current. Also output impedance of Wilson source is 0.5β times of the output impedance of simple current source which this issue significantly improves the performance of the circuit [6].

2. Q_{12} and Q_{13} push pull buffer and very large output impedance are separately used in order to improve the dynamic range of the amplifier [1].
3. D_2 and D_3 linear rectifiers are used in inputs of the differential amplifier, in order to decrease the distortion effect and increase the amplitude of the acceptable input voltage signals for linear performance of the amplifier [1].

4. Internal circuit of NE5517

NE5517 and LM13700 have a similar structure. Figure 6 shows the internal circuit of the 2 OTAs in this integrated circuit. The only difference between NE5517 and LM13700 is in the output high impedance buffer. Buffer in NE5517 consists of two Darlington transistors and a bias network that change the current of buffer bias according to IABC reference current. Therefore, changes in output offset voltages are almost eliminated .this issue is one of the advantages of NE5517 instead of LM13600 [3].

5. Internal circuit of CA3080

According to Figure 7, the internal circuit of CA3080 is presented [2]. General structure of this IC is similar to Figure 1 block diagram. Its current source is the same mirror current source, its active load 3 is Wilson current source, and its active loads 1 and 2 are the improved type of Wilson current source in which the transistor of output current source is used as Darlington form in that. Figure 8 shows the improved circuit of this current source without considering D_2 and D_4 diodes. These two diodes are employed in order to increase the speed of the source and they don't have any impact on general performance of the current source. They are assessed in 1-3 section. The only difference between Figure 8 and the circuit of Figure 5 of is the Darlington combination of Q_5 and Q_6, which is replaced by Q_7. Because the β of

Darlington combination is β^2 times of Q_7, output impedance of the current source will highly increase. In other words, the relation of Io output current with reference source is:



Figure 6. Internal circuit of NE5517



Figure 7. Internal circuit of CA3080

$$I_o = \frac{\beta^3 + 4\beta^2 + 4\beta}{\beta^3 + 3\beta^2 + 4\beta + 2} I_{ref} \tag{20}$$

As it is seen, by using an extra transistor, output current will have relation with the cube of β . therefore by the lowest value of βI_o becomes equal to I_{ref} and the impact of temperature on output current changes will significantly decreases.



Figure 8. The circuit of improved Wilson current source

6. The Model of Operational Transconductance Amplifier (OTA)

The proposed model for OTA consists of two major parts: Input differential level and output level. Input differential level is modelled with R_i input impedance, current source I_{ABC} based on $I_{ABC} = \frac{V_r - 2V_{BE} - (-V_{ee})}{R}$ equation for Wilson current supply and $I_{ABC} = \frac{V_r - V_{BE} - (\ V_{ee})}{R}$ equation for simple mirror current supply. V_r and R are the external factors applied to the IC by the designer. Figure 9 shows this basic section. Small resistors are employed in model in order to prevent the netlist and Convergence errors. n=1 coefficient for mirror current supply and n=2 for Wilson current supply are defined.



Figure 9. Equivalent circuit of OTA's Input level

In output level, active loads 2 and 3 transmit the current signal to output and supply output high impedance of OTA and are modelled as practical current source. This source consists of two parts. The first is the active load's output current signal modeller and second refers to its output impedance modeller. Current signal is equal to $i_{C2} = \frac{i_o}{2}$, or to half of the Equation (14), and is simply modelled as a current source dependant on current. But the output variable resistor of the source can't be directly defined based on the current of one component. Thus, the output impedance of Wilson active loads 2 and 3 that are $R_{O2} = \frac{1}{2}\beta\frac{V_A}{i_{C2}}$, can't be modelled in resistive form in order to be defined based on $i_{C2} = \frac{I_{ABC}}{2}$. On the other hand, a current source that is dependent on its voltage, in fact, is the modeller of a conductance element $(i = \frac{1}{R}V)$, so the output admittance of active load is modelled by dependent current source $\frac{I_{ABC}}{\beta V_A} = \frac{2i_{C2}}{\beta V_A}$. Figure 10, shows this equivalent circuit that contains:

$$G_2 = G_3 = \frac{1}{2} I_{ABC} \frac{e^{\frac{v_1 - v_2}{2V_T}} - e^{\frac{v_2 - v_1}{2V_T}}}{e^{\frac{v_1 - v_2}{2V_T}} + e^{\frac{v_2 - v_1}{2V_T}}} \quad (21)$$

$$GR_{O3} = GR_{O2} = \frac{I_{ABC}}{\beta V_A} \quad (22)$$

The written netlist for OTA covers the performance of Figure 9 and Figure 10 as below, that R_{in}, V_{BE}, n (equal to 1 or 2 for the definition of the current source), V_A and β are definable, and are easily determined by user.



Figure 10. OTA's equivalent circuit of output level

```
.Subckt OTA + - ABC V+ V- out  Params: n=2 VBE=0.65
+Rin=1e10 VA=75 B=100 VT=0.025
R1  +  1  0.01
R2  -  2  0.01
Ri  1  2  {Rin}               ; input impedence
R3  ABC  3  0.01
VABC  3  4  0                 ;  iABC sensor
V1  4  V-  {n*VBE} ;  2 VBE drift for vilson current source
Gro1 out  V+   value= {(I(VABC)*V(out,V+))/(2*B*VA)}          ;
+output Impedence Ro1=2Ro
Gro2 out  V-   value= {(I(VABC)*V(out,V-))/(2*B*VA)}          ;
+output Impedence Ro2=2Ro
G1  V+ out  value={0.5*I(VABC)*(exp(v(1,2)/(2*VT))-
+exp(v(2,1)/(2*VT)))/(exp(v(1,2)/(2*VT))+exp(v(2,1)/(2*VT)))}
G2  V-  out  value={0.5*I(VABC)*(exp(v(1,2)/(2*VT))-
+exp(v(2,1)/(2*VT)))/(exp(v(1,2)/(2*VT))+exp(v(2,1)/(2*VT)))}
.ENDS  OTA
```

7. The simulation of OTA's functional circuits

In this section, based on the presented model, three functional circuits that are introduced by the manufacturers will be simulated. OTA model has very suitable speed and accuracy, but it doesn't contain any characteristics of OTA such as frequency limitation, input offset currents and voltages .In order to show the accuracy of the model, these practical circuits once presented with the used model. Later on, they are simulated based on full internal circuit of LM13700 to understand the accuracy of the model's performance by comparing the results.

A. Voltage controlled Low pass filter

Figure 11 reflects the voltage controlled low pass filter [3].with respect to OTA's high input impedance and very small current in its terminals , the voltages on OTA's input terminals can be connected to input and output voltages as below.

$$v^{+} = V_3 = \frac{R_A}{R_A + R} V_{in} \tag{23}$$

$$v^{-} = V_4 = \frac{R_A}{R_A + R} V_{out} \tag{24}$$

Also output current of the OTA is completely crossed from the capacitor due to the presence of output buffer.



Figure 11. Voltage controlled low pass filter

$$v_C = g_m \left(v^{+} - v^{-}\right) \frac{1}{cs} = \frac{g_m R_A}{\left(R_A + R\right) cs} \left(V_{in} - V_{out}\right) \tag{25}$$

On the other hand, output Darlington transistors have common collector structure and employed as buffer. As a result, output signal V_{out} of this level is equal to input signal V_{in} , so:

$$V_{out} = \frac{g_m R_A}{\left(R_A + R\right) cs} \left(V_{in} - V_{out}\right) \tag{26}$$

After simplifying:

$$\frac{V_{out}}{V_{in}} = \frac{1}{1 + \frac{\left(R + R_A\right) c}{g_m R_A} s} \tag{27}$$

As it is obvious in Equation (27), Figure 11 is a linear low pass filter that, its cutoff frequency is placed in the pole's transfer function location. It means:

$$f = \frac{g_m \ R_A}{2\pi \left(R + R_A\right) C} \tag{28}$$

So, with gm gain coefficient controllable by potentiometer, the location of the filter's cutoff frequency can be simply changed. If IABC is adjusted on 0.75mA by using R_f potentiometer, then cutoff frequency of the filter is:

$$f = \frac{\frac{0.75154\,\text{m}}{2\times 25\text{m}}\times 200}{2\pi\left(100\text{k}+0.2\text{k}\right)\times 150\text{p}} = 29.475\text{kHz} \tag{29}$$

Figure 12 shows the used circuit for simulation of this low pass filter in PSpice environment with regard to represented model for OTA. The used netlist for the simulation of low pass filter and represented model are as below. The .lib command in this netlist shows the OTA's library path that the presented model is a script file that is stored by the name OTA.lib.

```
*Filter Schematics  Netlist With OTA model *
Q_Q1        Vcc   8  10  QY
Q_Q2        Vcc   5  8   QZ
V_V1        Vcc   0  12v
V_V2        Vee   0  -12v
R_R1        9     1  100k
R_RA        2     0  200
C_C         5     0  150p
R_RR        2     10 100k
R_R2        1     0  200
V_Vin       9     0  DC  0V  AC  1v
X_U1        1     2  6   Vcc Vee 5   OTA  PARAMS: n=2
+ VBE=0.7269   Rin=1e10   VA=200  B=200    VT=0.028
R_R3        7     6  2k
R_Rf        7     Vcc  {Rvar}
R_RL        10    Vee  10k

** Parameters, Subcircuits and Model Parameters **
.param Rvar=28k
.LIB "D:\OTA Paper\Pspice Lib\OTA.LIB"
.MODEL QY  NPN (IS=6E-15     BF=50)
.MODEL QZ  NPN (IS=5E-16     BF=266)

** Analysis setup **
.ac DEC 1000 1 100k
.OP
.PROBE V(*) I(*)
.PLOT ac VDB(10,9)  ; V(9)=Vin   V(10)=Vout

.END
```

The results of two mentioned netlists in simulation for 751.54µA are shown in Figure 13 and Figure 14. Table 1 compares the results of modelled simulation and internal circuit of LM13700 for 751.54µA control current with theoretical ones. As can be seen, the results of the two simulations are very close to each other, and its tolerance with the presented model is at last 4.1%.

Also the netlist of the simulation, with respect to the internal circuit of LM13700 is as below:

```
*Filter Schematics Netlist with internal circuit of LM13700 *
Q_HS1_Q2        13   6   15  QX1
Q_HS1_D4        18   18  Vcc QY1
Q_HS1_Q6        17   18  Vcc QY1
Q_HS1_Q7        20   17  18  QY1
Q_HS1_D6        22   22  Vcc QY1
Q_HS1_Q10       21   22  Vcc QY1
Q_HS1_Q4        17   2   13  QX1
Q_HS1_Q5        21   1   13  QX2
D_HS1_D2        23   2   Dx
D_HS1_D3        23   1   Dx
Q_HS1_D5        24   24  Vee QX1
Q_HS1_Q8        20   24  Vee QX1
Q_HS1_Q1        6    15  Vee QX1
Q_HS1_D1        15   15  Vee QX1
Q_HS1_Q11       5    21  22  QY1
Q_HS1_Q9        5    20  24  QX1
Q_HS1_Q12       Vcc  5   27  QZ
Q_HS1_Q13       Vcc  27  10  QY
V_V1            Vcc  0   12v
V_V2            Vee  0   -12v
R_R1            9    1   100k
R_R2            1    0   200
R_RA            2    0   200
R_RR            2    10  100k
R_RL            10   Vee 10k
R_R3            7    6   2k
C_C             5    0   150p
R_RF            7    Vcc {Rvar}
V_Vin           9    0   DC  0V  AC  1v
** Parameters, Subcircuits and Model Parameters **
.param Rvar=28k
.MODEL QY  NPN (IS=6E-15 BF=50)
.MODEL DX  D   (IS=5E-16)
.MODEL QX1 NPN (IS=5E-16 BF=200 NE=1.15 ISE=.63E-16 IKF=1E-2)
.MODEL QY1 PNP (IS=5E-16 BF=200 NE=1.15 ISE=.63E-16 IKF=1E-2)
.MODEL QX2 NPN (IS=5.125E-16 BF=200 NE=1.15 ISE=.63E-16 IKF=1E-2)
.MODEL QZ  NPN (IS=5E-16  BF=266)
** Analysis setup **
.ac DEC 1000 1 100k
.OP
.PROBE V(*) I(*)
.END
```



Figure 12. Linear low pass filter with given model

Table 1. Comparison of theory and simulation result for low pass filter

	Cut off frequency
theory	29.475 K Hz
Simulation with proposed model	28.249 K Hz
Simulation with internal circuit of LM13700	28.550 K Hz



Figure 13. The results of low pass filter's simulation with given model



Figure 14. The results of low pass filter's simulation with internal circuit of LM13700

B. Square and triangle wave oscillator

Figure 15, shows the circuit of square and triangle wave oscillator (3). In this circuit, due to the big difference between first OTA's input terminals voltage, capacitor is charged with constant current of I_C and also in linear form. By the enhancement of capacitor's voltage, negative terminal of second OTA increases and decreases the input voltage difference of second OTA. Consequently, this OTA will go to negative saturation state. Because this negative voltage is returned to positive terminal of OTA as a feedback, it will causes first OTA's output current to remain in its negative limit ,and capacitor starts to discharge with a negative slope . This continuous charge and discharge continues by reaching the $2V_T$ in second OTA's input terminals that is the performance range of the OTA in linear state. In [3] frequency is equal to $\frac{I_C}{4R_A C I_A}$. If the I_C current is set to 0.67mA, by changing the V_C voltage control, with respect to I_A it will be equal to:

$$I_A = \frac{V_{cc} - (-V_{ee}) - 2V_{BE}}{R} = 0.5627\,\text{mA} \tag{30}$$

Figure 16, shows the used circuit for simulating this oscillator in PSpice environment with respect to the given model of the OTA.



Figure 16. Square and triangle wave oscillator with given model



Figure 17. The results of oscillator's simulation with the given model

Output waves and the results of the netlist's simulation are shown in Figure 17. Convergence error happens for LM13700 internal circuit's simulation in PSpice. In Table 2, the results of the simulation are compared to theoretical results for 0.67mA control current.

Oscillator's netlist with given model is as below:

```
*Oscillator Schematics Netlist *
R_R3        Vcc   6    47k
Q_Q2        Vcc   4    8    QY
Q_Q4        Vcc   9    10   QY
V_V1        Vcc   0    15
V_V2        Vee   0    -15v
R_R2        10    Vee  10k
R_RA        7     0    5.1k
Q_Q3        Vcc   7    9    QZ
Q_Q1        Vcc   2    4    QZ
X_U15       7     8    6    Vcc   Vee   7   OTA  PARAMS: n=2
+ VBE=0.65  Rin=1e10   VA=150     B=200     VT=0.025
X_U14       7     0    3    Vcc   Vee   2   OTA  PARAMS: n=2
+ VBE=0.65  Rin=1e10   VA=150     B=200     VT=0.025
R_R4        8     Vee  20k
C_C         2     0    {Cvar} IC=0
R_R1        3     5    30k
V_V3        5     0    {Vc}

** Parameters, Subcircuits and Model Parameters **
.PARAM      Cvar=100n   Vc=6.4
.LIB "D:\OTA Paper\Pspice Lib\OTA.LIB"
.MODEL QY  NPN (IS=6E-15     BF=50)
.MODEL QZ  NPN (IS=5E-16     BF=266)
.OPTIONS ABSTOL=1pA
.OPTIONS RELTOL=0.1
.OP

** Analysis setup **
.tran 0.2m 4m 0 1u
.PROBE V(*) I(*)
.PLOT tran V(8) V(10)  ; V(8)=Vout1   V(10)=Vout2
.END
```

Table 2. The comparison between theoretical results and oscillator circuit's simulation

	Oscillation frequency
Theory	583.62 Hz
Simulation with proposed model	545.5 Hz
Simulation with internal circuit of LM13700	Convergence error

C. AM modulator

Figure 18 shows the circuit of AM modulator in Pspice environment based on the given model (4). Modulator's netlist is obtained in below:

The input carries sin wave with $V_{car} = V_c \sin \omega_c t$. Bias current $I_Q = \frac{V_c - V_{cc}}{R_C}$ is supplied by DC voltage source and input signal is defined as $I_m \sin \omega_m t = m I_Q \sin \omega_m t$. Therefore, carrier frequency ω_c , modulation frequency ω_m and $m = I_m / I_Q$ coefficient should be valid in $-1 < m < 1$ condition. With respect to OTA's performance, Opamp's output signal is



Figure 18. AM modulator with the given model

```
*Modulator Schematics Netlist *
R_R1            1    2    1k
R_R3            3    0    91
E_U2            7    0    VALUE {LIMIT(V(0,5)*1E6,-15V,+15V)}
X_U1            2    3    4  Vcc Vee 5 OTA PARAMS: n=2 VBE=0
+  Rin=1e10     VA=150  B=200  VT=0.025
R_RF            5    7    1k
C_CF            5    7    100p
R_R2            2    0    100
V_Vee           Vee 0    {Vee}
V_Vcar          1    0
+SIN    0   {Vc}  {Wc}  0  0  0
I_Iabc          4    0
+SIN    0 {m*IQ}  {Wm}  0  0  0
V_Vc            6    0    {Vee+Rc*IQ}
R_Rc            6    4    {Rc}
V_Vcc           Vcc 0    {Vcc}

** Parameters, Subcircuits and Model Parameters **
.PARAM          IQ=1m     m=0.6    Wm=1k
.PARAM          Vcc=10    Vee=-10
.PARAM          Vc=0.1m   Wc=25k   Rc=1k
.LIB "D:\OTA Paper\Pspice Lib\OTA.LIB"
.MODEL QY  NPN (IS=6E-15     BF=50)
.MODEL QZ  NPN (IS=5E-16     BF=266)

** Analysis setup **
.tran 2m 10m 7m 1u
.OP
.PROBE V(*) I(*)
.PLOT tran  v(7)  ; v(7)=V(out)
.END
```

$$i_o = \frac{I_{ABC}}{2V_T}(v^+ - v^-) \tag{31}$$

based on Figure 18 :

$$v^+ = \frac{R_2}{R_{1+}R_2} V_{car} \tag{32}$$

and

$$I_{ABC} = I_Q + I_m \sin \omega_m t = I_Q (1 + \frac{I_m}{I_Q} \sin \omega t) = I_Q (1 + m \sin \omega t) \tag{33}$$

so

$$i_o = \frac{I_Q}{2V_T} \times \frac{R_2}{R_1 + R_2} V_{car} (1 + m \sin \omega t) \tag{34}$$

If it is assumed that in the performance frequency of the circuit capacitor C is opened, then all i_o current crosses the R_F resistor (current in Opamp's input terminal) is zero, and we will have

$$V_o = \frac{I_Q R_F}{2V_T} \times \frac{R_2}{R_1 + R_2} V_{car} (1 + m \sin \omega t) \tag{35}$$

In modulator's circuit, R_1 and R_2 resistors are in the form of voltage divider, because carrier's signal amplitude not to increase the OTA's load, also $R_1 \parallel R_2$ resistor is small and under 100 ohm in order to integrate the resistors in OTA's input terminals. C_F capacitor is necessary for the accurate performance of the circuit in resonance frequencies. C_F impedance is decreased in resonance frequencies and would cause, R_F to decrease in Equation (35). Accordingly, output amplitude decreases and prevents the swing in circuit. By applying simulation to theoretical value of $m = 0.6$, $F_c = 25\,\text{kHz}$ carrier frequency, input frequency, $V_m = 0.1\,\text{mV}$ and current bias $I_Q = 1\,\text{mA}$, an output wave is modulated such as Figure 19 in PSpice environment. The results of the simulation are specified in the Figure 19. Table.3 shows a comparison between the results of simulation and theoretical values. As it is seen, the results of the simulation are very accurate, by assuming internal circuit of LM13700 and convergence error happens during the simulation of AM modulator which gives no results.



Figure 19. Am modulator's output with modulation coefficient m=0.6

Table 3. Comparison results n between theoretical and the simulation of Modulator's circuit

	V_m (mV)	f_m (kHz)	V_c (mV)	f_c (kHz)
Theory	0.3	1	0.1	25
Simulation with model	0.290 15	1	0.108	25
Simulation with internal circuit	Convergence error			

8. Conclusion

In this study, a new modelling approach for the operational transconductance amplifier is proposed. The model is based on OTA theoretical analysis which can facilitate the trend of education for the undergraduate students.

To evaluate the accuracy of the model, the circuits of low pass filter, square and triangle wave oscillator and also AM modulator are simulated based on presented model and a good agreement has been obtained by comparing the results of the proposed model with theoretical ones. As well, the internal circuit of LM13700 has been simulated and the result has shown that it doesn't converge in oscillator and modulator circuits unlike the proper convergence speed of the presented model.

According to achieved results, it can be proved that the presented model for OTA has proper convenience, accuracy, speed and convergence. The simple structure of this model and its compatibility with the theoretical concepts and internal circuits of the most famous ICs have caused the afore-mentioned model to be a proper tool for education of the undergraduate of electronics students.

9. Acknowledgment

This work was partially supported by Iranian Technical and Vocational University, Rasht branch.

10. References

[1] 1. G. Hwang, "A Tutorial Strategy Supporting System for Distance Learning On Computer Networks", *IEEE Transaction on Education*, Vol. 41, No. 4, 1998.

[2] R. C. Jaeger and T. N. Bolalock, "Microelectronic Circuits Design", Mc. Graw-Hill, Forth Edition, 2011, ISBN: 978-0-07-338045-2

[3] D. A. Neamen, "Electric Circuit Analysis and Design", Second Edition, McGraw-Hill, Forth Edition, 2010, ISBN: 978-0-07-338064-3.

[4] M. H. Rashid, "Microelectronic Circuits Analysis and Design", Cengage Learning Inc. Second Edition, 2011, ISBN-13: 978-0-495-66772-8.

[5] S. Franco, "Design with Operational Amplifier and Analog Integrated Circuits", McGraw-Hill, Third Edition, 2001, ISBN: 0-07-232084-2

[6] T. L. Floyd, "Electronics Fundamentals: Circuits, Devices and Applications", Prentice Hall, 7th Edition, 2006, ISBN-13: 978-0132197090.

[7] M. H. Rashid, "Introduction to PSpice Using ORCAD For Circuits and Electronics", Prentice Hall, Third Edition, 2003, ISBN-13: 978-0131019881.

[8] G. W. Roberts, and A. S. Sedra, "Spice", Oxford University Press, Second Edition, 1997, ISBN: 0-19-510842-6

[9] Q. Liu, and C. O. Nwankpa, "Applications of Operational Transconductance Amplifier in Power System Analog Emulation", *IEEE*, ISBN: 0-7803-8834-8, 2005

[10] G. J. Gdmez, S.H.K. Embabi, E. Sdnchez-Sinencio, and M. C. Lefebvre, "A Nonlinear Macromodel for CMOS OTAs", *IEEE*, ISBN: 0-7803-2570-2, 1995.

[11] Q. L. Sun, J. Y. Liu, and M. L. Liu, "An Improved Nonlinear Macromodel of OTA", *International Conference on Circuits and Systems*, June, China, 1991.

[12] Y. Wei and A. Doboli, “Library of Structural Analog Cell Macromodels for Design of Continuous-Time Reconfigurable Modulators”, *ISCAS*, *IEEE*, ISBN: 0-7803-9390-2, 2006..
[13] C. Plett, and M. A. Copeland, “A Study of Tuning for Continuous-Time Filters Using Macromodels”, *IEEE Transactions on Circuits and Systems-II: Analog and Digital Signal Processing*, No. 8, August, 1992.
[14] G. Oltean, “High Level Model for a Simple Transconductance Operational Amplifier”, EUROCON, Serbia & Montenegro, Belgrade, November 22-24, 2005.
[15] http://www.national.com/assests/en/tools/spice/LM13700.mod
[16] H. Biagi, R. M. Stitt, B. Baker and S. Baier, “Burr-Brown Spice Based Macromodels, REV.F”, Burr-Brown Application Bulletin, AB-020F, January, 1995.
[17] W. Jung, “Using the LTC Opamp Macromodel” Linear Technology, Application Note 48, AN 48-1, November, 1991.
[18] D. Riemer, “HA2500/02 Spice Operational Amplifier Macro-Model”, Harris Semiconductor, Application Note NM25001, October, 1996.
[19] S. Petrie and C. Hymowitz, “Simulating Circuits With SCRs”, Intusoft Newletter #24, February, 1992.
[20] N. Mohan, Undeland and Robbins, “Power Electronics: Converters, Applications and Design”, *John Wiley International Edition*,Third Edition, 2003, ISBN: 0-471-22693-9
[21] I. Batarseh, “Power Electronic Circuits”, John Wiley, First Edition, 2003, ISBN-13: 978-0471126621
[22] M. H. Rashid, “Power Electronics: Circuits, Devices and Applications”, Pearson/Prentice Hall, Third Edition, 2003, ISBN: 9780131011403.
[23] National semiconductor, "LM13700 Dual Operational Transconductance Amplifier with Linearizing Diode and Buffers", Application Note, February,1995.
[24] Harris Semiconductor, "Applications of the CA3080 High Performance Operational Transconducter Amplifier ", Application Note, AN6668.1, November 1996.
[25] Philips Semiconductor, "NE5417/5517A, Dual Operational Trasconductance Amplifier", Product Specification, August 1994.

Mohammad Reza Modabbernia received the B.S. degree in Electronics Engineering and M.S. degree in control Engineering from KNT, the University of Technology, Tehran, IRAN in 1995 and 1998 respectively. He is the staff member of Electronic group of Technical and vocational university, Rasht branch, Rasht, IRAN. His research interests include Robust Control, Nonlinear Control and Power Electronics.

Yekta Nazarpour received her junior college diploma in computer science from Moein community college of technology, Rasht, Iran. She is currently the student of Computer engineering of Islamic Azad University, Lahijan Branch. Her research interests include optimization and Artificial intelligence.

Seyedeh Shiva Nejati received her B.Sc. degree in electrical engineering from Guilan University, Iran, in 2005 and the M.Sc. degree in Electrical Engineering from Tabriz University, Iran, in 2008. Her research interests include optical fiber, optoelectronics and optical devices. She is university lecturer at higher education Institute of Sardar Jangal, Iran.

Syrus Rouhollahi Moghadam received his diploma in Electronics, B.A. and M.A. in English in 1985, 2000 and 2012 successively. He is the staff member of Humanity Faculty group of Technical and Vocational University, Rasht Branch, Rasht Iran. His research interests include Electronic Circuits and Translation.